# MAX3736 Input/Output Model

SPICE I/O Macromodels aid in understanding signal integrity issues in electronic systems. Most of Maxim's High Frequency/Fiber Communication ICs utilize input and output (I/O) circuits with Current Mode Logic (CML), Positive Emitter Coupled Logic (PECL), and Low Voltage Differential Signal (LVDS) formats to transfer data into and out of an IC. These models are based on simplified circuit expressions that may include replacement of active circuit elements with ideal controlled voltage and current sources. As such, simulation with macromodels should be treated as 'typical' performance and not relied upon as final proof-of-design. Use of macromodel descriptions is not a substitute for worst-case design analysis, nor for testing real circuits over temperature, supply, and other operating limits.

The output format is provided as ASCII text netlists suitable for generic SPICE. This format is compatible with most versions of SPICE such as PSPICE and HSPICE. Additional information is found in HFAN 6.1 *Input/Output Models for Maxim Fiber Component*s.

To extract the circuit netlists using the Adobe Acrobat Reader follow these instructions. Select the "Text Select Tool" by clicking the left mouse button on this icon of the menu bar (a capital T with a small dashed box to the lower right). Highlight the desired netlist text with the cursor. Use the copy command from the edit menu to capture the selected lines. Then paste the selected lines into a text file editor and save the file with an extension compatible with the simulator.

Revision A1, April 18, 2005.

# Input Model for the MAX3736




Figure 1. Input Model for signal IN of the MAX3736.

# MAX3736 Laser Driver Output Model




Figure 1. Output stage for the MAX3736 including a simplified package model.

# Notes:

The schematic of Figure 1 represent the input stage of the Maxim MAX3736 3.2Gbps SFP Laser Driver. The input circuit is shown with the signal inputs (IN+, IN-). However, the models are given in generic SPICE, which only accepts node names as numbers. As discussed in the application note the input signals are described as (2101, 2102). Similarly the output signals in Figure 2 are described as (2000, 2001). These models are only valid at 25°C. The bias currents for the internal circuitry are modeled by ideal current sources. This model is not compensated for variations in VCC, so VCC equal to 3.3V should be used.

**The Input Stage:** The input stage has the sub-circuit "INDRV".

**The INDRV Sub-circuit**: The input package connects to a differential pair. The signal continues through the input buffer from nodes 3 and 4 (positive and negative nodes respectively). The input voltage should fit within certain parameters. $V_{ID}$ should be in between $200mV_{P-P}$ to $2400mV_{P-P}$. $V_{INCM}$ should be .6*VCC. See Appendix A for the input netlist.

**The Output Stage:** The output stage is described by the sub-circuit DRV. It has two large output driving transistors, whose collectors are connected to the pads and ESD structures. In the actual bond diagram, there are two pads and two bondwires for each output. For die simulations, the sub-circuit DRV should be used without the OUTPKG sub-circuit.

The output modulation current is varied by changing the current IE in the DRV subcircuit. Presently it is set at 60mA. **The output model is preliminary**. The electrical eye is somewhat close to the electrical eye measured in the laboratory, but there are differences, especially at low values of modulation current IE. Figure 3 is the electrical eye measured in the lab at high modulation currents. Presently the model does not seem to work properly at 10mA and thus the model is limited from 20mA to 60mA. Putting a filter at approximately (0.75) of the data rate shows an eye which more closely resembles the optical eye.



Figure 3. Electrical Eye of theMAX3736 laser diode at a high modulation current.

**Text File Format:** This model is shipped in “pdf” format. Models and netlists can be copied to text format in the Acrobat Reader by holding the left mouse button on the “Text Select Tool.” Then the user can “select” what netlist and/or subcircuit with the mouse. Then use the copy command from the “edit” menu to capture the selected lines. These lines can then be “pasted” into the user’s text file.

# Appendix A: Input Netlist

```
* 3736 Input Model
**********************************************
.OPT ACCT NOMOD LIMPTS=10000
.TEMP 25
.OP
.TRAN 2P 3n
**********************************************

* Voltage Source
VCC 101 0 3.3V

**********************************************
*  Add input source here.
*  The source should connect to node 2101 (VINP)
*  and 2102 (VINN).
*  Example:
VINP 50 0 PULSE (2.28 1.655 0.04n 50p 50p 260p 625p)
VINN 60 0 PULSE (1.655 2.28 0.04n 50p 50p 260p 625p)

R1 2101 50 50
R2 2102 60 50
C1 2101  0 .1p
C2 2102  0 .1p
**********************************************

**********************************************
XINPKG 2101 2102 600 601 INPKG
XINDRV 600 601 101 INDRV
**********************************************

**********************************************
.SUBCKT INDRV 600 601 101

R3 600 1 50
R4 601 1 50
R5 101 1 16k
R6 1 0 24k
R7 101 3 2k
R8 101 4 2k

* Differential Pair
XQ1 4 600 2 0 N102M010_2
XQ2 3 601 2 0 N102M010_2

I1 2 0 .2mA

* ESD Diodes
XD1 600 101 0 DE0900
XD2 0 600 0 DE0900
XD3 601 101 0 DE0900
XD4 0 601 0 DE0900

* Pad Structures
```

```
XP1 600 0 PAD4SQ3P7
XP2 601 0 PAD4SQ3P7

.ENDS INDRV
************************************************

************************************************
.SUBCKT INPKG 2101 2102 600 601

RP1  600 2002 25M
RP2  601 2003 25M
LP1  2002 2101 .5665N
LP2  2003 2102 .5665N

.ENDS INPKG
************************************************

************************************************
* Transistor Model
.SUBCKT N102M010_2 1 2 3 21
CP1SUB 2 201 2.423F
RP1SUB 20 201 100K
CTRENCH 1 202 10.017F
RFIELDEPI 202 21 956.635
RREVERT 202 20 1G
CBL 10 20 1.380F
RSUB 20 21 309.654K
CWAFER 20 21 1.701F
CP1EPI 10 12 1.523F
CP1P2 12 3 1.326F
RBX 2 12 117.486 TC=2.423M
RCX 1 10 43.405 TC=2.815M,503.797N
RCI 10 11 10.851 TC=2.815M,503.797N
REX 13 3 39.913
QN 11 12 13 11 TX 2
.MODEL TX NPN( IS=1.102E-018 XTI=3 EG=1.120 BF=380 BR=12 XTB=0 VAF=66
+   VAR=2.500 NF=1.018 NR=1.020 NE=2 NC=1.560 IKF=2.425M IKR=68.900U
+   ISE=5.512E-019 ISC=0 RB=58.743 RBM=44.057 IRB=248.040U CJE=2.619F
+   MJE=463M VJE=1.100 FC=990M CJC=1.747F MJC=350M VJC=1 TF=1.320P TR=5N
+   XTF=2 VTF=1.200 ITF=8.957M PTF=5 KF=545.257N AF=2.150 )
.ENDS N102M010_2
************************************************

************************************************
* Pad Model
.SUBCKT PAD4SQ3P7 1 3
CPAD 1 10 67.534F
REPI 10 20 378.507 TC=4.800M,5U
CTRENCH 21 20 22.531F
CBL 21 20 1.413P
RX 20 21 1G
RS 3 21 5.416K
CWAFER 21 3 2.587F
.ENDS PAD4SQ3P7
************************************************
```

```
**********************************************
* Diode Model
.SUBCKT DE0900 1 2 21
CTRENCH 2 202 58.872F
RFIELDEPI 202 21 163.328
RREVERT 202 21 1G
CBL 4 5 78.946F
RSUB 5 21 26.725K
CWAFER 5 21 19.712F
CP1EPI 1 4 76.086F
DD 1 4 DCB
RS 4 2 5.368 TC=4.277M,4.217U
.MODEL DCB D( IS=7.920E-018 N=1.050 CJO=216F VJ=800M M=500M )
.ENDS DE0900
**********************************************

.PROBE
.END
```

# Appendix B: Output Model

```
INPUT - MAX3736 OUTPUT CIRCUIT
*
*
.OPT ACCT NOMOD NOPAGE RELTOL=0.001

.TEMP 25
.OP
.TRAN 5ps 1000ps

*The power supply is 3.3 Volts.
VCC 101 0 DC 3.3
RLOAD1 2001 101 15
RLOAD2 2002 101 15
CLOAD1 2002 2001 0.5pF
XPKG1 2001 2002, 4 5 OUTPKG

XCIROUT 4 5, 2 3 101 DRV
*
VINP 2 0 PULSE (1.0 0.50 20.0p 120.0p 120.0p 192.0p 625.0p)
VINN 3 0 PULSE (0.50 1.0 20.0p 120.0p 120.0p 192.0p 625.0p)

.SUBCKT DRV 2001 2002, 2 3 101
RB1 4 2 25
RB2 5 3 25
CC2001 2001 5 63F
CC2002 2002 6 63F
XQ3 2001 4 61 0 N102M066A
XQ4 2002 5 62 0 N102M066A
RaIE60MA 61 6 0.01
RbIE60MA 62 6 0.01
*******************************************************************************
* To adjust modulation current change IE from 20 to 60mA
IE 6 0 60mA
*******************************************************************************
XESD1 2001 101 0 DE0172
XESD2 0 2001 0 DE0172
XESD3 2002 101 0 DE0172
XESD4 0 2002 0 DE0172
XESDP1 2001 101 PAD
XESDP2 2001 101 PAD
XESDN1 2002 101 PAD
XESDN2 2002 101 PAD

.ENDS DRV
*

.SUBCKT OUTPKG 214 213 314 313
* Pad->Lead Frame side 213 Chip side 313
* Pad->Lead Frame side 214 Chip side 314
*resistors
R14 314 8 0.200
R13 313 9 0.200
```

```
*inductors
l14 214 8 0.43n
l13 213 9 0.43n
K13_14 l13 l14 0.15
*capacitors
C14 214 0 0.12p
C13 213 0 0.12p
C13_14 213 214 0.037p
.ENDS OUTPKG
*
.SUBCKT N102M066A 1 2 3 21
* count =24
CP1SUB 2 201 37.541F
RP1SUB 20 201 100K
CTRENCH 1 202 189.150F
RFIELDEPI 202 21 50.730
RREVERT 202 20 1G
CBL 10 20 47.478F
RSUB 20 21 14.109K
CWAFER 20 21 37.339F
CP1EPI 10 12 46.017F
CP1P2 12 3 53.073F
RBX 2 12 2.995 TC=1.934M
RCX 1 10 846.613M TC=2.640M,410.600N
RCI 10 11 211.653M TC=2.640M,410.600N
REX 13 3 529.375M
QN 11 12 13 11 TX 24
*XREPORT1 0 REPORTERL1N11
*XREPORT2 0 REPORTERL1N12
.MODEL TX NPN( IS=6.926E-018 XTI=3 EG=1.120 BF=380 BR=12 XTB=0 VAF=66
+   VAR=2.500 NF=1.018 NR=1.020 NE=2 NC=1.560 IKF=15.238M IKR=432.900U
+   ISE=3.463E-018 ISC=1.558E-030 RB=17.968 RBM=13.476 IRB=1.558M
+   CJE=16.207F MJE=463M VJE=1.100 FC=990M CJC=7.862F MJC=350M VJC=1
+   TF=1.320P TR=5N XTF=2 VTF=1.200 ITF=56.277M PTF=5 KF=4.742N AF=2.150 )
.ENDS N102M066A
*
*
.SUBCKT DE0172 1 2 21
CTRENCH 2 202 21.527F
RFIELDEPI 202 21 446.429
*RREVERT 202 20 1G
CBL 4 5 19.573F
RSUB 5 21 82.499K
CWAFER 5 21 6.386F
CP1EPI 1 4 13.706F
DD 1 4 DCB
RS 4 2 26.809
*.MODEL RMODTC0 RES( TC1=4.361M TC2=4.344U )
*extractelement REPORT1 0 REPORTER PARAMS: P0=0 P1=17.20 P2=1 P3=1 P4=1
*extractelement REPORT2 0 REPORTER PARAMS: P0=17.20 P1=4 P2=2.064E-017 P3=0
*extractelement +                          P4=0 P5=1
.MODEL DCB D( IS=1.514E-018 N=1.050 CJO=41.280F VJ=800M M=500M IBV=1M )
.ENDS DE0172
*
.SUBCKT PAD 1 3
CPAD 1 10 99.132F
```

```
REPI 10 20 396.030 TC=4.800M,5U
CTRENCH 21 20 22.531F
CBL 21 20 1.413P
RX 20 21 1G
RS 3 21 5.416K
CWAFER 21 3 2.587F
*XREPORT1 0 REPORTERL1N34
.ENDS PAD
*
.SUBCKT PAD4OCT3 1 3
CPAD 1 10 41.303F
REPI 10 20 526.050 TC=4.800M,5U
CTRENCH 21 20 19.684F
CBL 21 20 1.079P
RX 20 21 1G
RS 3 21 6.858K
CWAFER 21 3 1.975F
*XREPORT1 0 REPORTERL1N37
.ENDS PAD4OCT3
*

.PRINT TRAN V(2001) V(2002)
.PROBE

.END
```